Mechanical Vibration Analysis

Introduction

This worksheet demonstrates how Maple can be used in the engineering application, Mechanical Vibration Analysis.

```
> restart:
  with(plots):
```

Theory

The linear second order differential equation is expressed in classical form:

```
> gen1:=M*Diff(x(t),x,x)+C*Diff(x(t),x)+K*x(t)=f;
```

$$gen1 := M \frac{\partial^2}{\partial x^2} x(t) + C \frac{\partial}{\partial x} x(t) + K x(t) = f \tag{2.1}$$

This equation can also be expressed in the differential operator form for ease of use:

```
> gen2:=M*(D@@2)(x)(t)+C*(D)(x)(t)+K*x(t)=f;
```

$$gen2 := M D^{(2)}(x)(t) + C D(x)(t) + K x(t) = f \tag{2.2}$$

By algebraically manipulating the expression, the form traditionally used by engineers is derived:

```
> t1:=subs(f/M=F,expand(gen2/M));
```

$$t1 := D^{(2)}(x)(t) + \frac{C D(x)(t)}{M} + \frac{K x(t)}{M} = F \tag{2.3}$$

```
> t1:=algsubs(C/M=2*zeta*omega,t1):t1:=algsubs(K/M=omega^2,t1);
```

$$t1 := 2 D(x)(t) \zeta \omega + x(t) \omega^2 + D^{(2)}(x)(t) = F \tag{2.4}$$

This form includes the damping ratio zeta, the natural frequency omega, and the external forcing term F. Consider only free vibration by setting $F = 0$.

```
> gen3:=subs(F=0,t1);
```

$$gen3 := 2 D(x)(t) \zeta \omega + x(t) \omega^2 + D^{(2)}(x)(t) = 0 \tag{2.5}$$

Solution

```
> sol1:=dsolve({gen3,x(0)=P,D(x)(0)=V},x(t));
```

$$sol1 := x(t) = \frac{\left(\omega \sqrt{\zeta^2 - 1}\, P + \zeta \omega P + V\right) e^{\left(-\zeta + \sqrt{\zeta^2 - 1}\right) \omega t}}{2 \sqrt{\zeta^2 - 1}\, \omega} - \frac{\left(-\omega \sqrt{\zeta^2 - 1}\, P + \zeta \omega P + V\right) e^{-\left(\zeta + \sqrt{\zeta^2 - 1}\right) \omega t}}{2 \sqrt{\zeta^2 - 1}\, \omega} \tag{3.1}$$

Post-Processing

Undamped Response

Zeta = 0:

```
> trial[1]:=rhs(subs(zeta=0,omega=5,P=5,V=0,sol1));
```

$$trial_1 := \frac{5\,e^{5\,I\,t}}{2} + \frac{5\,e^{-5\,I\,t}}{2} \qquad \textbf{(4.1.1)}$$

```
> plot1:=plot(trial[1], t=0..5, color=blue, legend =
  "Undamped"): display({plot1});
```



Under-damped Response

0 < Zeta < 1

```
> trial[2]:=rhs(subs(zeta=.2,omega=5,P=5,V=0,sol1));
```

$$trial_2 := (2.500000000 - 0.5103103630\,I)\,e^{(-1.0 + 4.898979486\,I)\,t} + (2.500000000 + 0.5103103630\,I)\,e^{(-1.0 - 4.898979486\,I)\,t} \qquad \textbf{(4.2.1)}$$

```
> plot2:=plot(trial[2], t=0..5, color=green, legend = "Under-damped"): display(plot2);
```



Critically Damped Response

Zeta = 1

```
> trial[3]:=rhs(dsolve({subs(zeta=1,omega=5,gen3),x(0)=P,D(x)(0)=V},x(t)));
```

$$trial_3 := P\,\mathrm{e}^{-5\,t} + (V + 5\,P)\,\mathrm{e}^{-5\,t}\,t \qquad \textbf{(4.3.1)}$$

```
> pl3:=subs(P=5,V=0,trial[3]);
```

$$pl3 := 5\,\mathrm{e}^{-5\,t} + 25\,\mathrm{e}^{-5\,t}\,t \qquad \textbf{(4.3.2)}$$

```
> plot3:=plot(pl3, t=0..5, color=cyan, legend = "Critically Damped"): display(plot3);
```



Over-Damped Response

$1 < \text{Zeta}$

```
> trial[4]:=rhs(subs(zeta=1.3, omega=5,P=5,V=0,sol1));
```

$$trial_4 := 6.412540225\ \mathrm{e}^{-2.346688068\,t} - 1.412540225\ \mathrm{e}^{-10.65331193\,t} \tag{4.4.1}$$

```
> plot4:=plot(trial[4],t=0..5, color=red, legend = "Over-
  Damped"): display(plot4);
```



Comparison Plot

```
> display({plot1,plot2,plot3,plot4},title = "Response
  Comparison", size=[800,400]);
```


Response Comparison
4
2
0
-2
-4
1
2
3
4
5
t
Undamped
Under-damped
Critically Damped
Over-Damped